
1. [bookmark: _Toc122434485][bookmark: _Toc137814840]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 22.1.2 which is part of the NBIOT test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc137814841]Verification Test Summary 
Test Case: 	22.1.2
ATS Version:	iwd-TTCN3-B2022-09_D23wk24
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	MediaTek Dimensity D930
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc137814842]Corrections required
1. [bookmark: _Toc137814843][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	f_NBIOT_InitSystemInformation()

	Reason for change
	For system information combination c8 and c9, the new value of SIB2 including ntn_ConfigCommon_r1 is not updated in v_SI_List

	Summary of change
	Update v_SI_List[0] with modified SIB2 contents

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



Before change
  function f_NBIOT_InitSystemInformation(NBIOT_CellId_Type                         p_CellId,
                                         EUTRA_ASN1_CellIdentity_Type              p_CellIdentity,
                                         NB_Combination_Type                       p_Combination,
                                         integer                                   p_FrequencyBand,
                                         NB_OperationMode_Type                     p_OperationMode,
                                         template (value) PLMN_IdentityList_NB_r13 p_PLMN_Identity,
                                         EUTRA_ASN1_TrackingAreaCode_Type          p_TrackingAreaCode,
                                         EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN)
     runs on NBIOT_PTC return template (value) NB_BcchInfo_Type
  {
    var template (value) NB_BcchInfo_Type v_CellBcchInfo;
    var template (value) BCCH_BCH_Message_NB v_MIB;
    var template (value) BCCH_DL_SCH_Message_NB v_SIB1;
    var template (value) NB_SI_List_Type v_SI_List;
    var template (value) SystemInformation_NB_r13_IEs v_SIB2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI3;
    var template (omit) NRS_CRS_PowerOffset_Type v_NRS_CRS_PowerOffset;
    var template (omit) NB_EutraControlRegionSize_Type v_NB_EutraControlRegionSize;
    var template (value) NB_OperationModeInfo_Type v_OperationMode;
    var template (value) SchedulingInfoList_NB_r13 v_SchedulingInfoList;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI2;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI3;
    var NB_CarrierFreq_Type v_CarrierFreq := fl_NBIOT_Get_CarrierFreq(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN); // @sic R5-185113 sic@

    select (p_OperationMode) {
      case (standalone) {
        v_OperationMode := cs_NB_OperationModeInfo_Standalone;
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (inband_samePCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandSamePCI (fl_NBIOT_MIB_GetEUTRA_CRS_SequenceInfo(p_FrequencyBand)); // @sic R5-181295 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := dB0;
      }
      case (inband_differentPCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandDifferentPCI (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (guardband) {
        v_OperationMode := cs_NB_OperationModeInfo_GuardBand (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
    }
    // --- Initialise MIB ---
    v_MIB :=  cs_508_MasterInformationBlock_NB_Init(v_OperationMode);
    
    // --- Initialise SIB1 ---
    // -- A default scheduling assigned ----------
    v_SIB1 := cs_508_SystemInformationBlockType1_NB_Def(p_PLMN_Identity,
                                                        p_TrackingAreaCode,
                                                        p_CellIdentity,
                                                        p_FrequencyBand,
                                                        v_NB_EutraControlRegionSize,
                                                        v_NRS_CRS_PowerOffset,
                                                        cs_SchedulingInfo_NB_1SI);
    // --- Initialise SIB2 ---
    v_SIB2 := cs_SI_NB_SIB2(f_NBIOT_InitSystemInformation2(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN));
    v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2);
    
    // --- Initialise all SYSINFO --
    if (p_Combination == c1) {
      v_SchedulingInfoList := cs_SchedulingInfo_NB_1SI;
    } else if (p_Combination == c5) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB20(cs_508_SystemInformationBlockType20_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType20_NB_r14});
    } else if (p_Combination == c6) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType22_NB_r14});
    } else if (p_Combination == c7) { //@sic R5-2226052 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SI_List[2] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB23(cs_508_SystemInformationBlockType23_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType22_NB_r14}, -, -, -, {});
      v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.schedulingInfoList_v1530 := cs_SchedulingInfo_NB_c7_r15;
    } else if (p_Combination == c8 or p_Combination == c9) { //@sic R5-231904, R5-231933, R5-233367, R5-233365 sic@
        v_SI_List[1] := f_NBIOT_InitSIB31(p_CellId);
        v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({});
        v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
        v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
        v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
        if (p_Combination == c8){
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode);
        }
        else{//c9 (c8+SIB3)
            v_SI_List[2] := v_SI_List[1];
            v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def));
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType3_NB_r13});
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType3_NB_r13}, -, -, -, {});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c9_r15);
        }
    } else {
      // --- Initialise SIB3 ---
      v_SIB_MappingInfo_SI2 := {sibType3_NB_r13};
      v_SIB_SI2 := cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def);
      
      select (p_Combination) { // @sic R5-185113 sic@
        case (c2) { // SIB3 & SIB4 Intra-Frequency
          v_SIB_MappingInfo_SI3 := {sibType4_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB4(cs_508_SystemInformationBlockType4_NB_Def);
        }
        case (c3) { // SIB 3 & SIB5 Inter-Frequency
          v_SIB_MappingInfo_SI3 := {sibType5_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB5(f_NBIOT_InitSystemInformation5(p_CellId, p_FrequencyBand, p_OperationMode));
        }
        case (c4) { // SIB 3 & 14
          v_SIB_MappingInfo_SI3 := {sibType14_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB14(cs_SystemInformationBlockType14_NB_Def);
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid sysinfo combination");
        }
      }
      v_SI_List[1] := cs_SystemInformation_Common_NB(v_SIB_SI2);
      v_SI_List[2] := cs_SystemInformation_Common_NB(v_SIB_SI3);
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, v_SIB_MappingInfo_SI2, -, -, -, v_SIB_MappingInfo_SI3);
    }
    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    return v_CellBcchInfo;
  }
  

After change
  function f_NBIOT_InitSystemInformation(NBIOT_CellId_Type                         p_CellId,
                                         EUTRA_ASN1_CellIdentity_Type              p_CellIdentity,
                                         NB_Combination_Type                       p_Combination,
                                         integer                                   p_FrequencyBand,
                                         NB_OperationMode_Type                     p_OperationMode,
                                         template (value) PLMN_IdentityList_NB_r13 p_PLMN_Identity,
                                         EUTRA_ASN1_TrackingAreaCode_Type          p_TrackingAreaCode,
                                         EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN)
     runs on NBIOT_PTC return template (value) NB_BcchInfo_Type
  {
    var template (value) NB_BcchInfo_Type v_CellBcchInfo;
    var template (value) BCCH_BCH_Message_NB v_MIB;
    var template (value) BCCH_DL_SCH_Message_NB v_SIB1;
    var template (value) NB_SI_List_Type v_SI_List;
    var template (value) SystemInformation_NB_r13_IEs v_SIB2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI3;
    var template (omit) NRS_CRS_PowerOffset_Type v_NRS_CRS_PowerOffset;
    var template (omit) NB_EutraControlRegionSize_Type v_NB_EutraControlRegionSize;
    var template (value) NB_OperationModeInfo_Type v_OperationMode;
    var template (value) SchedulingInfoList_NB_r13 v_SchedulingInfoList;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI2;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI3;
    var NB_CarrierFreq_Type v_CarrierFreq := fl_NBIOT_Get_CarrierFreq(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN); // @sic R5-185113 sic@

    select (p_OperationMode) {
      case (standalone) {
        v_OperationMode := cs_NB_OperationModeInfo_Standalone;
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (inband_samePCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandSamePCI (fl_NBIOT_MIB_GetEUTRA_CRS_SequenceInfo(p_FrequencyBand)); // @sic R5-181295 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := dB0;
      }
      case (inband_differentPCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandDifferentPCI (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (guardband) {
        v_OperationMode := cs_NB_OperationModeInfo_GuardBand (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
    }
    // --- Initialise MIB ---
    v_MIB :=  cs_508_MasterInformationBlock_NB_Init(v_OperationMode);
    
    // --- Initialise SIB1 ---
    // -- A default scheduling assigned ----------
    v_SIB1 := cs_508_SystemInformationBlockType1_NB_Def(p_PLMN_Identity,
                                                        p_TrackingAreaCode,
                                                        p_CellIdentity,
                                                        p_FrequencyBand,
                                                        v_NB_EutraControlRegionSize,
                                                        v_NRS_CRS_PowerOffset,
                                                        cs_SchedulingInfo_NB_1SI);
    // --- Initialise SIB2 ---
    v_SIB2 := cs_SI_NB_SIB2(f_NBIOT_InitSystemInformation2(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN));
    v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2);
    
    // --- Initialise all SYSINFO --
    if (p_Combination == c1) {
      v_SchedulingInfoList := cs_SchedulingInfo_NB_1SI;
    } else if (p_Combination == c5) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB20(cs_508_SystemInformationBlockType20_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType20_NB_r14});
    } else if (p_Combination == c6) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType22_NB_r14});
    } else if (p_Combination == c7) { //@sic R5-2226052 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SI_List[2] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB23(cs_508_SystemInformationBlockType23_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType22_NB_r14}, -, -, -, {});
      v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.schedulingInfoList_v1530 := cs_SchedulingInfo_NB_c7_r15;
    } else if (p_Combination == c8 or p_Combination == c9) { //@sic R5-231904, R5-231933, R5-233367, R5-233365 sic@
        v_SI_List[1] := f_NBIOT_InitSIB31(p_CellId);
        v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({});
        v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
        v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
        v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
         v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2);
        if (p_Combination == c8){
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode);
        }
        else{//c9 (c8+SIB3)
            v_SI_List[2] := v_SI_List[1];
            v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def));
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType3_NB_r13});
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType3_NB_r13}, -, -, -, {});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c9_r15);
        }
    } else {
      // --- Initialise SIB3 ---
      v_SIB_MappingInfo_SI2 := {sibType3_NB_r13};
      v_SIB_SI2 := cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def);
      
      select (p_Combination) { // @sic R5-185113 sic@
        case (c2) { // SIB3 & SIB4 Intra-Frequency
          v_SIB_MappingInfo_SI3 := {sibType4_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB4(cs_508_SystemInformationBlockType4_NB_Def);
        }
        case (c3) { // SIB 3 & SIB5 Inter-Frequency
          v_SIB_MappingInfo_SI3 := {sibType5_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB5(f_NBIOT_InitSystemInformation5(p_CellId, p_FrequencyBand, p_OperationMode));
        }
        case (c4) { // SIB 3 & 14
          v_SIB_MappingInfo_SI3 := {sibType14_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB14(cs_SystemInformationBlockType14_NB_Def);
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid sysinfo combination");
        }
      }
      v_SI_List[1] := cs_SystemInformation_Common_NB(v_SIB_SI2);
      v_SI_List[2] := cs_SystemInformation_Common_NB(v_SIB_SI3);
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, v_SIB_MappingInfo_SI2, -, -, -, v_SIB_MappingInfo_SI3);
    }
    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    return v_CellBcchInfo;
  }
  
  

1. [bookmark: _Toc137814844]Change 2
	Function name
	f_NBIOT_InitSIB31()

	Reason for change
	The array vc_NBIOT_Global.CellArray() is not initialized until the function f_NBIOT_CellArray_Init() returns (in function fl_NBIOT_Common_Init()) so it cannot be referenced in this function which is called multiple times during f_NBIOT_CellArray_Init()

	Summary of change
	Remove check on length of vc_NBIOT_Global.CellArray

	TTCN module
	NBIoT_CellInfo.ttcn

	MCC160 Comment
	



Before change
  
  function f_NBIOT_InitSIB31(NBIOT_CellId_Type p_CellId,
                               template (omit) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17 := omit)
    runs on NBIOT_PTC return template (value) BCCH_DL_SCH_Message_NB
    {
        var template (value) SystemInformationBlockType31_NB_r17 v_SIB31 := cs_508_SystemInformationBlockType31_NB_GSO_Def;
        var integer v_Index := f_NBIOT_CellInfo_GetIndex(p_CellId);
        if(lengthof(vc_NBIOT_Global.CellArray)==lengthof(cs_SystemInformationBlockType31_NB_r17_Array))
        {
            if(f_NBIOT_MobileInfo_GetNTNSatelliteType() == ntn_GSO)
            {
                v_SIB31 :=cs_SystemInformationBlockType31_NB_r17_Array[v_Index];
                if(isvalue(p_UL_SyncValidityDuration_r17))
                {
                    v_SIB31.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17  //compiler warning may be ignored due to isvalue check
                }
            }
            else
            {
                FatalError(__FILE__, __LINE__, "NGSO configuration is not yet supported");
            }
        }
        else
        {
            FatalError(__FILE__, __LINE__, "SIB 31 configuration for this NBIOT cell is not yet supported");
        }
        
        return cs_SystemInformation_Common_NB(cs_SI_NB_SIB31(v_SIB31));
        
    }

After change
  function f_NBIOT_InitSIB31(NBIOT_CellId_Type p_CellId,
                               template (omit) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17 := omit)
    runs on NBIOT_PTC return template (value) BCCH_DL_SCH_Message_NB
    {
        var template (value) SystemInformationBlockType31_NB_r17 v_SIB31 := cs_508_SystemInformationBlockType31_NB_GSO_Def;
        var integer v_Index := f_NBIOT_CellInfo_GetIndex(p_CellId);
        /*REMOVED if(lengthof(vc_NBIOT_Global.CellArray)==lengthof(cs_SystemInformationBlockType31_NB_r17_Array))
        {*/
            if(f_NBIOT_MobileInfo_GetNTNSatelliteType() == ntn_GSO)
            {
                v_SIB31 :=cs_SystemInformationBlockType31_NB_r17_Array[v_Index];
                if(isvalue(p_UL_SyncValidityDuration_r17))
                {
                    v_SIB31.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17  //compiler warning may be ignored due to isvalue check
                }
            }
            else
            {
                FatalError(__FILE__, __LINE__, "NGSO configuration is not yet supported");
            }
        /* REMOVED }
        else
       {
            FatalError(__FILE__, __LINE__, "SIB 31 configuration for this NBIOT cell is not yet supported");
        }*/
        
        return cs_SystemInformation_Common_NB(cs_SI_NB_SIB31(v_SIB31));
        
    }
  
  

1. [bookmark: _Toc137814845]Change 3
	Function name
	f_TC_22_1_2_NBIOT()

	Reason for change
	UL grant transmission is not started

	Summary of change
	Start default UL grant transmission after receiving RRCConnectionRequest 

	TTCN module
	NBIoT_General_NTN.ttcn

	MCC160 Comment
	



Before change
  
  function f_TC_22_1_2_NBIOT() runs on NBIOT_PTC
  { // NB-IoT / NTN
    var NAS_MSG_Indication_Type v_NasInd;
//    var template (present) NAS_UL_Message_Type v_AttachReq;
    var template (present) NAS_UL_Message_Type v_ESMMessage;
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (present) UL_DCCH_Message_NB v_RRCConnectionSetupComplete_NB := cr_508_RRCConnectionSetupComplete_NB;

    //Initialise all cell, security and mobile parameters
    f_NBIOT_Init(c8);

    //Create and configure NCell 1
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //UE is Switched ON to configure Automatic PLMN selection mode
    f_UT_SwitchOnUE(UT, false);
    //Make sure the UE is in automatic PLMN selection mode
    f_UT_AutomaticPLMN_Select (UT);
    //Make sure that No UE location is pre-configured
    f_UT_ConfigureNTNPosition(UT,"No UE location is pre-configured");
    //Now Power OFF the UE
    f_UT_PowerOffUE (UT);
    f_NBIOT_TestBody_Set(true);

    //Now bring up the cell to serving cell power for premable
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);

    //@siclog "Step 1" siclog@
    f_UT_SwitchOnUE(UT, false);
    
    //@siclog "Step 1A" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 11
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell1, 360.0)) {
      //Does the UE sent RRCConnectionRequest-NB message on Ncell 1 within 6 minutes
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 1A : RRCConnectionRequest-NB message received on NCell 1 within 6 minutes");
    }
    //@siclog "Step 1B siclog@
    f_UT_PowerOffUE (UT);
    //@siclog "Step 1B siclog@
    f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 1C siclog@
    f_UT_ConfigureNTNPosition(UT);
    //@siclog "Step 2 siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell1, 360.0)) {
      //Does the UE sent RRCConnectionRequest-NB message on Ncell 1 within 6 minutes
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 2 : RRCConnectionRequest-NB message received on NCell 1 within 6 minutes");
    }
    
    //@siclog "Steps 3 - 14 siclog@
    //Steps 3 to 14 of the registration procedure described in TS 36.508 [18] subclause 8.1.5.2.3 are performed on Ncell 1.
    f_NBIOT_RRC_ConnectionSetup_Def();

    if (px_DoAttachWithoutPDN) {
      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;
    } else {
      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType);
    }
    //The rrcConnectionSetupComlete should include IE gnss-ValidityDuration-r17, but contents are not checked according to Table 22.1.2.3.3-1
    v_RRCConnectionSetupComplete_NB.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.gnss_ValidityDuration_r17 := ?;
    
    v_NasInd := f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell1,
                                                       v_RRCConnectionSetupComplete_NB,
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                  cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT),
                                                                                  v_ESMMessage));
    
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);

    //@siclog "Step 15" siclog@
    // The SS transmits an RRCConnectionRelease-NB message to release RRC connection and move to RRC_IDLE
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);
    f_NBIOT_TestBody_Set(false);
    
    //Switch/power off UE
    f_NBIOT_Postamble (nbiot_Cell1, CONTROL_PLANE, N1_IDLE);
  }

After change
  
  function f_TC_22_1_2_NBIOT() runs on NBIOT_PTC
  { // NB-IoT / NTN
    var NAS_MSG_Indication_Type v_NasInd;
//    var template (present) NAS_UL_Message_Type v_AttachReq;
    var template (present) NAS_UL_Message_Type v_ESMMessage;
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (present) UL_DCCH_Message_NB v_RRCConnectionSetupComplete_NB := cr_508_RRCConnectionSetupComplete_NB;
  var template (omit) integer v_HarqProcessNumber;

    //Initialise all cell, security and mobile parameters
    f_NBIOT_Init(c8);

    //Create and configure NCell 1
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //UE is Switched ON to configure Automatic PLMN selection mode
    f_UT_SwitchOnUE(UT, false);
    //Make sure the UE is in automatic PLMN selection mode
    f_UT_AutomaticPLMN_Select (UT);
    //Make sure that No UE location is pre-configured
    f_UT_ConfigureNTNPosition(UT,"No UE location is pre-configured");
    //Now Power OFF the UE
    f_UT_PowerOffUE (UT);
    f_NBIOT_TestBody_Set(true);

    //Now bring up the cell to serving cell power for premable
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);

    //@siclog "Step 1" siclog@
    f_UT_SwitchOnUE(UT, false);
    
    //@siclog "Step 1A" siclog@
    //Check for no RRCConnectionRequest-NB message on Ncell 11
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell1, 360.0)) {
      //Does the UE sent RRCConnectionRequest-NB message on Ncell 1 within 6 minutes
      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 1A : RRCConnectionRequest-NB message received on NCell 1 within 6 minutes");
    }
    //@siclog "Step 1B siclog@
    f_UT_PowerOffUE (UT);
    //@siclog "Step 1B siclog@
    f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 1C siclog@
    f_UT_ConfigureNTNPosition(UT);
    //@siclog "Step 2 siclog@
    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell1, 360.0)) {
      //Does the UE sent RRCConnectionRequest-NB message on Ncell 1 within 6 minutes
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 2 : RRCConnectionRequest-NB message received on NCell 1 within 6 minutes");
    }
    
    v_HarqProcessNumber := f_NBIOT_CellInfo_GetDefaultHarqProcessNumber(nbiot_Cell1);
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant(cs_NB_DciInfo_CcchDcchDtchUL(-, -, -, v_HarqProcessNumber))));
    //@siclog "Steps 3 - 14 siclog@
    //Steps 3 to 14 of the registration procedure described in TS 36.508 [18] subclause 8.1.5.2.3 are performed on Ncell 1.
    f_NBIOT_RRC_ConnectionSetup_Def();

    if (px_DoAttachWithoutPDN) {
      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;
    } else {
      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType);
    }
    //The rrcConnectionSetupComlete should include IE gnss-ValidityDuration-r17, but contents are not checked according to Table 22.1.2.3.3-1
    v_RRCConnectionSetupComplete_NB.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.gnss_ValidityDuration_r17 := ?;
    
    v_NasInd := f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell1,
                                                       v_RRCConnectionSetupComplete_NB,
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                  cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT),
                                                                                  v_ESMMessage));
    
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);

    //@siclog "Step 15" siclog@
    // The SS transmits an RRCConnectionRelease-NB message to release RRC connection and move to RRC_IDLE
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);
    f_NBIOT_TestBody_Set(false);
    
    //Switch/power off UE
    f_NBIOT_Postamble (nbiot_Cell1, CONTROL_PLANE, N1_IDLE);
  }	
  


1. [bookmark: _Toc137814846]Change 4
	Function name
	f_NBIOT_RRC_RRCConnectionRequest()

	Reason for change
	RRCConnectionRequest with Initial UE identity as “randomValue” should be allowed

	Summary of change
	Allow any value for initial UE identity

	TTCN module
	NBIOT_RRC_Templates.ttcn

	MCC160 Comment
	



Before change
  
  function f_NBIOT_RRC_RRCConnectionRequest_Check(template (present) NBIOT_CellId_Type p_CellId,
                                                  float p_TimerValue,
                                                  template (present) EstablishmentCause_NB_r13 p_EstablishmentCause := cr_EstablishmentCause_NB_Any,
                                                  boolean p_JustCheck := false)
    runs on NBIOT_PTC return boolean
  {
    var template NB_SRB_COMMON_IND v_SRB0_RrcPdu_IND := car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionRequest_NB(p_EstablishmentCause));
    
    timer t_Wait;
    t_Wait.start(p_TimerValue);
    alt {
      // receive RRC Connection Request on SRB0 (CCCH):
      [not p_JustCheck] SRB.receive(v_SRB0_RrcPdu_IND)
        {
          t_Wait.stop;
          return true;
        }
      [p_JustCheck] SRB.check(receive(v_SRB0_RrcPdu_IND))
        {
          t_Wait.stop;
          return true;
        }
      [] t_Wait.timeout
        {
        }
    }
    return false;
  }

After change
  
  	function f_NBIOT_RRC_RRCConnectionRequest_Check(template (present) NBIOT_CellId_Type p_CellId,
                                                  float p_TimerValue,
                                                  template (present) EstablishmentCause_NB_r13 p_EstablishmentCause := cr_EstablishmentCause_NB_Any,
                                                  boolean p_JustCheck := false)
    runs on NBIOT_PTC return boolean
  {
    var template NB_SRB_COMMON_IND v_SRB0_RrcPdu_IND := car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionRequest_NB(p_EstablishmentCause, ?));
    
    timer t_Wait;
    t_Wait.start(p_TimerValue);
    alt {
      // receive RRC Connection Request on SRB0 (CCCH):
      [not p_JustCheck] SRB.receive(v_SRB0_RrcPdu_IND)
        {
          t_Wait.stop;
          return true;
        }
      [p_JustCheck] SRB.check(receive(v_SRB0_RrcPdu_IND))
        {
          t_Wait.stop;
          return true;
        }
      [] t_Wait.timeout
        {
        }
    }
    return false;
  }
  



1. [bookmark: _Toc137814847]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NB-IoT band 255 using ciphering algorithm eea1 and integrity algorithm eia1 
3 [bookmark: _Toc137814848]Execution Log Files
3.1 [bookmark: _Toc137814849][bookmark: _Hlk120026754]MediaTek Dimensity D930
The MediaTek Dimensity D930 passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_22_1_2_LOG.html
· PICS/PIXIT parameter file:
       TC_22_1_2_PIXIT.xml


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc137814850]References 
	[1]
	R5s230479 : Supporting information for addition of Rel-17 NB-IoT NTN test case 22.1.2      



